MODELS OF BIOLOGICAL INTERACTION

Please refer to the Word document "Models of Biological Interaction Among Species" for a full written commentary

Interaction models

Model of competition between species, with no crowding or self-limiting effects (page 3 of Word file)

```
> restart;with(DEtools):
> with(linalg):
```

```
> Feq5:=(4-3*P)*F;
```

$$Feq5 := (4-3\,P)\,F$$

```
> Peq6:=(3-F)*P;
```

$$Peq6 := (3-F)\,P$$

```
> sol1:=solve({Feq5=0,Peq6=0},{F,P});
```

$$sol1 := \left\{F=3, P=\frac{4}{3}\right\}, \{F=0, P=0\}$$

```
> sol11:=sol1[1];
```

$$sol11 := \left\{F=3, P=\frac{4}{3}\right\}$$

```
> sol12:=sol1[2];
```

$$sol12 := \{F=0, P=0\}$$

Obtaining the phase diagram for the model:

Note
The following procedure for plotting multiple trajectories along with the direction field is adapted from Commbes, K., R. et al (2nd ed. 1997) Differential Equations with Maple. John Wiley & Sons Inc., Chapter 12.

```
> des:=diff(F(t),t)=4*F(t)-3*F(t)*P(t),diff(P(t),t)=3*P(t)-F(t)*P
  (t):
  iniset:={seq(seq([F(0)=a,P(0)=b],a=[0.5,1.5,2.5,3.5]), b=[0.5,
  1.5,2,2.5])}:
  pphase:=trange->DEplot([des],[F(t),P(t)],t=trange,iniset,F=0..6,
  P=0..4,stepsize=.05,      method=rkf45,linecolour=black,arrows=
  thin):
```

```
> pphase(-2..3);
```



Which shows a saddle point around the fixed point

$(F, P) = \left(3, \frac{4}{3}\right)$

Predator-Prey Models (Lotka-Volterra)

With no self-limitations on species/population size (no crowding) in either population (see Page 5 of Word file)

```
> restart: with(plots):  with(DEtools):
> dFdt1:=diff(F(t),t)=(a-b*P(t))*F(t);
```

$$dFdt1 := \frac{d}{dt} F(t) = (a - b\, P(t))\, F(t)$$

```
> dPdt1:=diff(P(t),t)=((-1)*c + d*F(t))*P(t);
```

$$dPdt1 := \frac{d}{dt} P(t) = (-c + d\, F(t))\, P(t)$$

```
> dFdt2:=subs({a=1,b=1,c=0.1, d=0.1},dFdt1);
```

$$dFdt2 := \frac{d}{dt} F(t) = (1 - P(t))\, F(t)$$

```
> dPdt2:=subs({a=1,b=1,c=0.1, d=0.1},dPdt1);
```

$$dPdt2 := \frac{d}{dt} P(t) = (-0.1 + 0.1\, F(t))\, P(t)$$

```
> dFdt:=rhs(dFdt2); dPdt:=rhs(dPdt2);
```

$$dFdt := (1 - P(t))\, F(t)$$

$$dPdt := (-0.1 + 0.1\, F(t))\, P(t)$$

```
> init_con3 := {F(0)=2,P(0)=1};
```

$$init\_con3 := \{F(0) = 2, P(0) = 1\}$$

```
> des:=diff(F(t),t)=dFdt,diff(P(t),t)=dPdt;
```

$$des := \frac{d}{dt} F(t) = (1 - P(t))\, F(t), \frac{d}{dt} P(t) = (-0.1 + 0.1\, F(t))\, P(t)$$

```
> iniset:={seq(seq([F(0)=alpha,P(0)=beta],alpha=[2,4.9]), beta=[1,
  1])};
```

$$iniset := \{[F(0) = 2, P(0) = 1], [F(0) = 4.9, P(0) = 1]\}$$

```
> pphase:=trange->DEplot([des],[F(t),P(t)],t=trange,iniset,F=0..5,
  P=0..2,stepsize=.05,method=rkf45,linecolour=black,arrows=thin):
> pphase(-4..22);
```




```
> rate_eqn1:= diff(F(t),t)=dFdt; rate_eqn2:=diff(P(t),t)=dPdt;
  vars:= [F(t), P(t)];
```

$$rate\_eqn1 := \frac{d}{dt} F(t) = (1 - P(t))\, F(t)$$

$$rate\_eqn2 := \frac{d}{dt} P(t) = (-0.1 + 0.1\, F(t))\, P(t)$$

$$vars := [F(t), P(t)]$$

```
> init1:=[F(0)=2,P(0)=1]; init2:=[F(0)=4.9,P(0)=1]; domain := 0 ..
  50;
```

$$init1 := [F(0) = 2, P(0) = 1]$$

$$init2 := [F(0) = 4.9, P(0) = 1]$$

$$domain := 0 ..50$$

We now plot the predator (P) and prey (F) populations jointly against time using the first of the given initial conditions. You should repeat this with the other initial conditions. Get a feeling for the accuracy of the computations by changing the step size, and for the long term behavior by changing the time interval.

```
> L:= DEplot({rate_eqn1, rate_eqn2}, vars, domain,{init1 },
  linecolor=blue, stepsize=0.5, scene=[t, F(t)], arrows=NONE):
  H:= DEplot({rate_eqn1, rate_eqn2}, vars, domain,{init1 },
  linecolor=red,stepsize=0.5, scene=[t, P(t)], arrows=NONE):
> display( {L,H} , title = `Predators (P, red) and Prey (F, blue)
  vs. time` );
```



We now plot the predator (P) and prey (F) populations jointly against time using the second of the given initial conditions.

```
> L2:= DEplot({rate_eqn1, rate_eqn2}, vars, domain,{init2 },
  linecolor= green, stepsize=0.5, scene=[t, F(t)], arrows=NONE):
  H2:= DEplot({rate_eqn1, rate_eqn2}, vars, domain,{init2 },
  linecolor=black,stepsize=0.5, scene=[t, P(t)], arrows=NONE):
> display( {L2, H2} , title = `Predators (P, green) and Prey (F,
  black) vs. time` );
```



Although it is not done here, you should get a feeling for the accuracy of the computations by changing the step size, and for the long term behavior by changing the time interval.

Next we plot the predator and prey populations against one another in a PHASE PORTRAI, using a different (and simpler) Maple command than used previously in this file. We do this for two different initial conditions. [Can you identify which curve goes with which initial condition? How is the independent variable t showing up in these pictures? (Hint: try it again with time interval t = 0 .. 20.)]

```
> DEplot({rate_eqn1, rate_eqn2}, vars, t= 0 .. 160, {init1, init2}
  , stepsize=0.5, scene=[F,P],linecolor=blue,title=`Predators (P)
  vs. Prey (F) for t = 0 .. 160`, arrows=slim);
```



As you play with the models, keep these questions in mind:

1. What is the long term behavior of the system?
2. In the case of oscillations, what is the period (time interval from peak to peak or trough to trough), and what is the amplitude?
3. How does changing the initial conditions affect your answers to questions 1 and 2?
4. Does the system have any steady states (equilibria)? Do these appear to be stable or unstable?
5. If there are steady states, are they in any way related to the long term behavior?

What is the significance of the next calculation? (Hint: try using these values of F and P as initial conditions.)

```
> rhs1:=rhs(rate_eqn1);rhs2:=rhs(rate_eqn2);
```

$$rhs1 := (1 - P(t))\, F(t)$$

$$rhs2 := (-0.1 + 0.1\, F(t))\, P(t)$$

```
> rhs11:=subs({P(t)=P, F(t)=F},rhs1);rhs22:=subs({P(t)=P, F(t)=F},
  rhs2);
```

$$rhs11 := (1 - P)\, F$$

$$rhs22 := (-0.1 + 0.1\, F)\, P$$

```
> equil:= solve( {rhs11, rhs22}, {F , P });
```

$$equil := \{F = 1., P = 1.\}, \{F = 0., P = 0.\}$$

Next we study solutions of the Lotka-Volterra system where the prey is assumed to grow logistically in the absence of any predators. (see page 7 of Word file)

```
> restart: with(plots):  with(DEtools):
> dFdt1:=diff(F(t),t)=(a-b*P(t)-u*F(t))*F(t);
```

$$dFdt1 := \frac{d}{dt} F(t) = (a - b\, P(t) - u\, F(t))\, F(t)$$

```
> dPdt1:=diff(P(t),t)=((-1)*c + d*F(t))*P(t);
```

$$dPdt1 := \frac{d}{dt} P(t) = (-c + d\, F(t))\, P(t)$$

```
> rhsF:=rhs(dFdt1);rhsP:=rhs(dPdt1);
```

$$rhsF := (a - b\, P(t) - u\, F(t))\, F(t)$$

$$rhsP := (-c + d\, F(t))\, P(t)$$

```
> solutions1:=solve({rhsF,rhsP},{P(t),F(t)});
```

$$solutions1 := \left\{F(t) = \frac{c}{d}, P(t) = \frac{a\, d - u\, c}{b\, d}\right\}, \left\{F(t) = \frac{a}{u}, P(t) = 0\right\}, \{F(t) = 0, P(t) = 0\}$$

```
> dFdt2:=subs({a=1,b=1,u=0.1, c=0.1, d=0.1},dFdt1);
```

$$dFdt2 := \frac{d}{dt} F(t) = (1 - P(t) - 0.1\, F(t))\, F(t)$$

```
> dPdt2:=subs({a=1,b=1,u=0.1, c=0.1,d=0.1},dPdt1);
```

$$dPdt2 := \frac{d}{dt} P(t) = (-0.1 + 0.1\, F(t))\, P(t)$$

```
> dFdt:=rhs(dFdt2); dPdt:=rhs(dPdt2);
```

$$dFdt := (1 - P(t) - 0.1\, F(t))\, F(t)$$

$$dPdt := (-0.1 + 0.1\, F(t))\, P(t)$$

```
> rate_eq1:= diff(F(t),t)=dFdt; rate_eq2:=diff(P(t),t)=dPdt;
  vars:= [F(t), P(t)];
```

$$rate\_eq1 := \frac{d}{dt} F(t) = (1 - P(t) - 0.1\, F(t))\, F(t)$$

$$rate\_eq2 := \frac{d}{dt} P(t) = (-0.1 + 0.1\, F(t))\, P(t)$$

$$vars := [F(t), P(t)]$$

```
> init1:=[F(0)=3,P(0)=0.1]; init2:=[F(0)=3,P(0)=1.5]; domain := 0
  .. 100;
```

$$init1 := [F(0) = 3, P(0) = 0.1]$$

$$init2 := [F(0) = 3, P(0) = 1.5]$$

$$domain := 0 \,..\, 100$$

We plot the predator and prey populations jointly against time using the first of the given initial conditions. You should repeat this with the other initial conditions. Get a feeling for the accuracy of the computations by changing the step size, and for the long term behavior by changing the time interval.

```
> L:= DEplot({rate_eq1, rate_eq2}, vars, domain,{init1 },
  linecolor=blue, stepsize=0.5, scene=[t, F(t)], arrows=NONE):
  H:= DEplot({rate_eq1, rate_eq2}, vars, domain,{init1 },
  linecolor=red,stepsize=0.5, scene=[t, P(t)], arrows=NONE):
```

```
> display( {L,H} , title = `Predators (P, red) and Prey (F, blue)
  vs. time: logistic prey` );
```




```
> DEplot({rate_eq1, rate_eq2}, vars, t= 0 .. 160, {init1, init2},
  stepsize=0.5, scene=[F,P],linecolor=blue,title=`Predators (P)
  vs. Prey (F) for t = 0 .. 160: Logistic prey`, arrows=slim);
```



As you play with the models, keep these questions in mind:

1. What is the long term behavior of the system?
2. In the case of oscillations, what is the period (time interval from peak to peak or trough to trough), and what is the amplitude?
3. How does changing the initial conditions affect your answers to questions 1 and 2?
4. Does the system have any steady states (equilibria)? Do these appear to be stable or unstable?
5. If there are steady states, are they in any way related to the long term behavior?

What is the significance of the next calculation? (Hint: try using these values of F and C as initial conditions.)

```
> rhs1:=rhs(rate_eq1);rhs2:=rhs(rate_eq2);
```

$$rhs1 := (1 - P(t) - 0.1\,F(t))\,F(t)$$

$$rhs2 := (-0.1 + 0.1\,F(t))\,P(t)$$

```
> rhs11:=subs({P(t)=P, F(t)=F},rhs1);rhs22:=subs({P(t)=P, F(t)=F},
  rhs2);
```

$$rhs11 := (1 - P - 0.1\,F)\,F$$

$$rhs22 := (-0.1 + 0.1\,F)\,P$$

```
> equil:= solve( {rhs11, rhs22}, {F , P });
```

$$equil := \{F = 1., P = 0.9000000000\}, \{F = 10., P = 0.\}, \{F = 0., P = 0.\}$$

What is the significance of this last calculation? ***Answer questions 1-5 for this model.***

Next we study solutions of the Lotka-Volterra system where the prey and predator populations both grow logistically.

```
> restart: with(plots):  with(DEtools):
> dFdt1:=diff(F(t),t)=(a-b*P(t)-u*F(t))*F(t);
```

$$dFdt1 := \frac{d}{dt} F(t) = (a - b\,P(t) - u\,F(t))\,F(t)$$

```
> dPdt1:=diff(P(t),t)=(c-v*P(t)+d*F(t))*P(t);
```

$$dPdt1 := \frac{d}{dt} P(t) = (c - v\,P(t) + d\,F(t))\,P(t)$$

```
> dFdt:=rhs(dFdt1);
```

$$dFdt := (a - b\,P(t) - u\,F(t))\,F(t)$$

```
> dPdt:=rhs(dPdt1);
```

$$dPdt := (c - v\,P(t) + d\,F(t))\,P(t)$$

```
> dFdt2:=subs({a=1,b=1,u=1/10, c=1/10, d=1/10, v=1/8},dFdt);
```

$$dFdt2 := \left(1 - P(t) - \frac{1}{10} F(t)\right) F(t)$$

```
> dPdt2:=subs({a=1,b=1,u=1/10, c=1/10, d=1/10,v=1/8},dPdt);
```

$$dPdt2 := \left(\frac{1}{10} - \frac{1}{8} P(t) + \frac{1}{10} F(t)\right) P(t)$$

```
> rate_eq1:= diff(F(t),t)=dFdt2; rate_eq2:=diff(P(t),t)=dPdt2;
  vars:= [F(t), P(t)];
```

$$rate\_eq1 := \frac{d}{dt} F(t) = \left(1 - P(t) - \frac{1}{10} F(t)\right) F(t)$$

$$rate\_eq2 := \frac{d}{dt} P(t) = \left(\frac{1}{10} - \frac{1}{8} P(t) + \frac{1}{10} F(t)\right) P(t)$$

$$vars := [F(t), P(t)]$$

```
> init1:=[F(0)=2,P(0)=1.4]; init2:=[F(0)=2,P(0)=1.2]; domain := 0
  .. 100;
```

$$init1 := [F(0) = 2, P(0) = 1.4]$$

$$init2 := [F(0) = 2, P(0) = 1.2]$$

$$domain := 0..100$$

We plot the predator and prey populations jointly against time using the first of the given initial conditions. You should repeat this with the other initial conditions. Get a feeling for the accuracy of the computations by changing the step size, and for the long term behavior by changing the time interval.

```
> L:= DEplot({rate_eq1, rate_eq2}, vars, domain,{init1 },
  linecolor=blue, stepsize=0.5, scene=[t, F(t)], arrows=NONE):
  H:= DEplot({rate_eq1, rate_eq2}, vars, domain,{init1 },
  linecolor=red,stepsize=0.5, scene=[t, P(t)], arrows=NONE):
> display( {L,H} , title = `Predators (P, red) and Prey (F, blue)
  vs. time: logistic prey` );
```




```
> DEplot({rate_eq1, rate_eq2}, vars, t= 0 .. 160, {init1, init2},
  stepsize=0.5, scene=[F,P],linecolor=blue,title=`Predators (P)
  vs. Prey (F) for t = 0 .. 160: Logistic P and F`, arrows=slim);
```



As you play with the models, keep these questions in mind:

1. What is the long term behavior of the system?
2. In the case of oscillations, what is the period (time interval from peak to peak or trough to trough), and what is the amplitude?
3. How does changing the initial conditions affect your answers to questions 1 and 2?
4. Does the system have any steady states (equilibria)? Do these appear to be stable or unstable?
5. If there are steady states, are they in any way related to the long term behavior?

What is the significance of the next calculation? (Hint: try using these values of F and C as initial conditions.)

```
> rhs1:=rhs(rate_eq1);rhs2:=rhs(rate_eq2);
```

$$rhs1 := \left(1 - P(t) - \frac{1}{10}\,F(t)\right)F(t)$$

$$rhs2 := \left(\frac{1}{10} - \frac{1}{8}\,P(t) + \frac{1}{10}\,F(t)\right)P(t)$$

```
> rhs11:=subs({P(t)=P, F(t)=F},rhs1);rhs22:=subs({P(t)=P, F(t)=F},
  rhs2);
```

$$rhs11 := \left(1 - P - \frac{1}{10} F\right) F$$

$$rhs22 := \left(\frac{1}{10} - \frac{1}{8} P + \frac{1}{10} F\right) P$$

```
> equilibrium1:=solve({dFdt=0,dPdt=0},{F(t), P(t)});
```

$$equilibrium1 := \left\{F(t) = \frac{a\,v - b\,c}{u\,v + d\,b}, P(t) = \frac{u\,c + d\,a}{u\,v + d\,b}\right\}, \left\{F(t) = 0, P(t) = \frac{c}{v}\right\}, \left\{F(t) = \frac{a}{u}, P(t) = 0\right\}, \{F(t) = 0, P(t) = 0\}$$

```
> equil:= solve( {rhs11, rhs22}, {F , P });
```

$$equil := \left\{F = \frac{2}{9}, P = \frac{44}{45}\right\}, \left\{F = 0, P = \frac{4}{5}\right\}, \{F = 10, P = 0\}, \{F = 0, P = 0\}$$

The Conrad 2-species interaction model.

```
> restart: with(plots):  with(DEtools):
```

Herbivore (H) dynamics:

```
> dHdt1:=diff(H(t),t)=h*H(t)*(1-(H(t)/(HMAX)))-beta*H(t)*P(t);
```

$$dHdt1 := \frac{d}{dt} H(t) = h\,H(t) \left(1 - \frac{H(t)}{HMAX}\right) - \beta\,H(t)\,P(t)$$

Predator (P) dynamics:

```
> dPdt1:=diff(P(t),t)=p*P(t)*(1-(P(t)/PMAX))+chi*H(t)*P(t);
```

$$dPdt1 := \frac{d}{dt} P(t) = p\,P(t) \left(1 - \frac{P(t)}{PMAX}\right) + \chi\,H(t)\,P(t)$$

```
> dHdt:=rhs(dHdt1);
```

$$dHdt := h\,H(t) \left(1 - \frac{H(t)}{HMAX}\right) - \beta\,H(t)\,P(t)$$

```
> dPdt:=rhs(dPdt1);
```

$$dPdt := p\,P(t) \left(1 - \frac{P(t)}{PMAX}\right) + \chi\,H(t)\,P(t)$$

```
> dHdt2:=subs({g=1.5, C=0, beta=0.001, chi=0.00001, HMAX=10000, h=
  0.5, p=0.2,PMAX=250},dHdt);
```

$$dHdt2 := 0.5\,H(t) \left(1 - \frac{1}{10000} H(t)\right) - 0.001\,H(t)\,P(t)$$

```
> dPdt2:=subs({g=1.5, C=0, beta=0.001, chi=0.00001, HMAX=10000, h=
  0.5, p=0.2,PMAX=250},dPdt);
```

$$dPdt2 := 0.2\,P(t) \left(1 - \frac{1}{250} P(t)\right) + 0.00001\,H(t)\,P(t)$$

```
> rate_eq1:= diff(H(t),t)=dHdt2; rate_eq2:=diff(P(t),t)=dPdt2;
  vars:= [ H(t), P(t)];
```

$$rate\_eq1 := \frac{d}{dt}\, H(t) = 0.5\, H(t) \left(1 - \frac{1}{10000}\, H(t)\right) - 0.001\, H(t)\, P(t)$$

$$rate\_eq2 := \frac{d}{dt}\, P(t) = 0.2\, P(t) \left(1 - \frac{1}{250}\, P(t)\right) + 0.00001\, H(t)\, P(t)$$

$$vars := [H(t), P(t)]$$

```
> init1:=[H(0)=10000,P(0)=250]; init2:=[H(0)=4000,P(0)=400];
  domain := 0 .. 100;
```

$$init1 := [H(0) = 10000, P(0) = 250]$$

$$init2 := [H(0) = 4000, P(0) = 400]$$

$$domain := 0\,..100$$

We plot the predator and prey populations jointly against time using the second of the given initial conditions. You should repeat this with the other initial conditions. Get a feeling for the accuracy of the computations by changing the step size, and for the long term behavior by changing the time interval.

```
> Z:= DEplot({rate_eq1, rate_eq2}, vars, domain,{init2 },
  linecolor=black, stepsize=0.5, scene=[t, H(t)], arrows=slim):
```

```
> L:= DEplot({rate_eq1, rate_eq2}, vars, domain,{init2 },
  linecolor=blue, stepsize=0.5, scene=[t, P(t)], arrows=NONE):
```

```
> display( {Z} , title = `Herbivores (H, black) vs. time` );
  display( {L} , title = `Predators (P, blue) vs. time` );
```



Herbivores (H, black) vs. time
4000
3900
3800
3700
3600
3500
0
20
40
60
80
100
t




```
> DEplot({rate_eq1, rate_eq2}, vars, t= 0 .. 160, {init1, init2},
  stepsize=0.5, scene=[H,P],linecolor=blue,title=`Predators (P)
  vs. Prey (H) for t = 0 .. 160: Logistic prey`, arrows=slim);
```



As you play with the models, keep these questions in mind:

1. What is the long term behavior of the system?
2. In the case of oscillations, what is the period (time interval from peak to peak or trough to trough), and what is the amplitude?
3. How does changing the initial conditions affect your answers to questions 1 and 2?
4. Does the system have any steady states (equilibria)? Do these appear to be stable or unstable?
5. If there are steady states, are they in any way related to the long term behavior?

What is the significance of the next calculation? (Hint: try using these solution values of F and C as initial conditions.)

```
> rhs1:=rhs(rate_eq1);rhs2:=rhs(rate_eq2);
```

$$rhs1 := 0.5\, H(t) \left(1 - \frac{1}{10000}\, H(t)\right) - 0.001\, H(t)\, P(t)$$

$$rhs2 := 0.2\, P(t) \left(1 - \frac{1}{250}\, P(t)\right) + 0.00001\, H(t)\, P(t)$$

```
> rhs11:=subs({P(t)=P, H(t)=H},rhs1);rhs22:=subs({P(t)=P, H(t)=H},
  rhs2);
```

$$rhs11 := 0.5\,H\left(1 - \frac{1}{10000}\,H\right) - 0.001\,H\,P$$

$$rhs22 := 0.2\,P\left(1 - \frac{1}{250}\,P\right) + 0.00001\,H\,P$$

```
> equil:= solve( {rhs11, rhs22}, {H , P });
```

$$equil := \{H = 0., P = 0.\}, \{H = 0., P = 250.\}, \{H = 10000., P = 0.\}, \{H = 4000., P = 300.\}$$

The Conrad 4-species interaction model (grass, herbivore and predator + cattle).

```
> restart: with(plots):  with(DEtools):
```

Grass (G) dynamics (with cattle, C, a fixed number):

```
> dGdt1:=diff(G(t),t)=g*G(t)*(1-(G(t)/GMAX))-alpha[1]*H(t)-alpha
  [2]*C;
```

$$dGdt1 := \frac{d}{dt}\,G(t) = g\,G(t)\left(1 - \frac{G(t)}{GMAX}\right) - \alpha_1\,H(t) - \alpha_2\,C$$

Herbivore (H) dynamics:

```
> dHdt1:=diff(H(t),t)=h*H(t)*(1-(H(t)/(theta*G(t))))-beta*H(t)*P
  (t);
```

$$dHdt1 := \frac{d}{dt}\,H(t) = h\,H(t)\left(1 - \frac{H(t)}{\theta\,G(t)}\right) - \beta\,H(t)\,P(t)$$

Predator (P) dynamics:

```
> dPdt1:=diff(P(t),t)=p*P(t)*(1-(P(t)/PMAX))+chi*H(t)*P(t);
```

$$dPdt1 := \frac{d}{dt}\,P(t) = p\,P(t)\left(1 - \frac{P(t)}{PMAX}\right) + \chi\,H(t)\,P(t)$$

```
> dGdt:=rhs(dGdt1);
```

$$dGdt := g\,G(t)\left(1 - \frac{G(t)}{GMAX}\right) - \alpha_1\,H(t) - \alpha_2\,C$$

```
> dHdt:=rhs(dHdt1);
```

$$dHdt := h\,H(t)\left(1 - \frac{H(t)}{\theta\,G(t)}\right) - \beta\,H(t)\,P(t)$$

```
> dPdt:=rhs(dPdt1);
```

$$dPdt := p\,P(t)\left(1 - \frac{P(t)}{PMAX}\right) + \chi\,H(t)\,P(t)$$

```
> dGdt2:=subs({g=1.5,alpha[1]=20,alpha[2]=200, C=0, beta=0.001,
  chi=0.00001, theta=0.01, h=0.5, p=0.2,PMAX=250,GMAX=1000000},
  dGdt);
```

$$dGdt2 := 1.5\, G(t) \left(1 - \frac{1}{1000000}\, G(t)\right) - 20\, H(t)$$

```
> dHdt2:=subs({g=1.5,alpha[1]=20,alpha[2]=200, C=0, beta=0.001,
  chi=0.00001, theta=0.01, h=0.5, p=0.2,PMAX=250,GMAX=1000000},
  dHdt);
```

$$dHdt2 := 0.5\, H(t) \left(1 - \frac{100.\, H(t)}{G(t)}\right) - 0.001\, H(t)\, P(t)$$

```
> dPdt2:=subs({g=1.5,alpha[1]=20,alpha[2]=200, C=0, beta=0.001,
  chi=0.00001, theta=0.01, h=0.5, p=0.2,PMAX=250,GMAX=1000000},
  dPdt);
```

$$dPdt2 := 0.2\, P(t) \left(1 - \frac{1}{250}\, P(t)\right) + 0.00001\, H(t)\, P(t)$$

```
> rate_eq0:= diff(G(t),t)=dGdt2; rate_eq1:= diff(H(t),t)=dHdt2;
  rate_eq2:=diff(P(t),t)=dPdt2;
  vars:= [G(t), H(t), P(t)];
```

$$rate\_eq0 := \frac{d}{dt}\, G(t) = 1.5\, G(t) \left(1 - \frac{1}{1000000}\, G(t)\right) - 20\, H(t)$$

$$rate\_eq1 := \frac{d}{dt}\, H(t) = 0.5\, H(t) \left(1 - \frac{100.\, H(t)}{G(t)}\right) - 0.001\, H(t)\, P(t)$$

$$rate\_eq2 := \frac{d}{dt}\, P(t) = 0.2\, P(t) \left(1 - \frac{1}{250}\, P(t)\right) + 0.00001\, H(t)\, P(t)$$

$$vars := [G(t), H(t), P(t)]$$

```
> init1:=[G(0)=500000, H(0)=3000,P(0)=200]; init2:=[G(0)=900000, H
  (0)=9000,P(0)=200]; domain := 0 .. 100;
```

$$init1 := [G(0) = 500000, H(0) = 3000, P(0) = 200]$$

$$init2 := [G(0) = 900000, H(0) = 9000, P(0) = 200]$$

$$domain := 0\,..100$$

We plot the grass, predator and prey populations jointly against time using the first of the given initial conditions. You should repeat this with the other initial conditions. Get a feeling for the accuracy of the computations by changing the step size, and for the long term behavior by changing the time interval.

```
> Z:= DEplot({rate_eq0, rate_eq1, rate_eq2}, vars, domain,{init1 }
  , linecolor=black, stepsize=0.5, scene=[t, G(t)], arrows=NONE):
```

```
> L:= DEplot({rate_eq0, rate_eq1, rate_eq2}, vars, domain,{init1 }
  , linecolor=blue, stepsize=0.5, scene=[t, H(t)], arrows=NONE):
  F:= DEplot({rate_eq0, rate_eq1, rate_eq2}, vars, domain,{init1 }
  , linecolor=red,stepsize=0.5, scene=[t, P(t)], arrows=NONE):
```

```
> display( {Z} , title = `Grass (G, black) vs. time: all species
```

```
have logistic growth` );display( {L} , title = `Herbivore Prey
(H, blue) vs. time: all species have logistic growth` );display(
{F} , title = `Predators (P, red) vs. time: all species have
logistic growth` );display( {L,F} , title = `Predators (P, red)
and Herbivore Prey (H, blue) vs. time: all species have logistic
growth` );
```




Herbivore Prey (H, blue) vs. time: all species have logistic growth
4000
3800
3600
3400
3200
3000
0
20
40
60
80
100
t


Predators (P, red) vs. time: all species have logistic growth
290
280
270
260
250
240
230
220
210
200
0
20
40
60
80
100
t




```
> FF:= DEplot({rate_eq0, rate_eq1, rate_eq2}, vars, domain,{init1,
  init2 }, linecolor=blue,stepsize=0.5, scene=[G(t), H(t)],
  arrows=SLIM):
> display( {FF} , title = `Grass (G) and Herbivore Prey (H): all
  species have logistic growth` );
```




```
> DEplot({rate_eq0,rate_eq1, rate_eq2}, vars, t= 0 .. 160, {init1,
  init2}, stepsize=0.5, scene=[H,P],linecolor=blue,title=
  `Predators (P) vs. Prey (H) for t = 0 .. 160: Logistic prey`,
  arrows=slim);
```



As you play with the models, keep these questions in mind:

1. What is the long term behavior of the system?
2. In the case of oscillations, what is the period (time interval from peak to peak or trough to trough), and what is the amplitude?
3. How does changing the initial conditions affect your answers to questions 1 and 2?
4. Does the system have any steady states (equilibria)? Do these appear to be stable or unstable?
5. If there are steady states, are they in any way related to the long term behavior?

What is the significance of the next calculation? (Hint: try using these values of F and C as initial conditions.)

```
> rhs0:=rhs(rate_eq0);rhs1:=rhs(rate_eq1);rhs2:=rhs(rate_eq2);
```

$$rhs0 := 1.5\, G(t) \left(1 - \frac{1}{1000000}\, G(t)\right) - 20\, H(t)$$

$$rhs1 := 0.5\, H(t) \left(1 - \frac{100.\, H(t)}{G(t)}\right) - 0.001\, H(t)\, P(t)$$

$$rhs2 := 0.2\, P(t) \left(1 - \frac{1}{250}\, P(t)\right) + 0.00001\, H(t)\, P(t)$$

```
> rhs00:=subs({P(t)=P, H(t)=H, G(t)=G},rhs0);rhs11:=subs({P(t)=P, H(t)=H, G(t)=G},rhs1);rhs22:=subs({P(t)=P, H(t)=H, G(t)=G},rhs2)
;
```

$$rhs00 := 1.5\, G \left(1 - \frac{1}{1000000}\, G\right) - 20\, H$$

$$rhs11 := 0.5\, H \left(1 - \frac{100.\, H}{G}\right) - 0.001\, H\, P$$

$$rhs22 := 0.2\, P \left(1 - \frac{1}{250}\, P\right) + 0.00001\, H\, P$$

```
> equil:= solve( {rhs00=0, rhs11=0, rhs22=0}, {G, H , P });
```

$$equil := \{G = 1.000000\ 10^{6}, H = 0., P = 0.\}, \{G = 8.666666667\ 10^{5}, H = 8666.666667, P = 0.\}, \{G = 1.000000\ 10^{6}, H = 0., P = 250.\}, \{G = 9.460853079\ 10^{5}, H = 3825.592358, P = 297.8199045\}, \{G = -3.946085308\ 10^{6}, H = -1.463825592\ 10^{6}, P = -18047.81990\}$$

END OF DOCUMENT